Multivariate Stirling Polynomials

Tutorial and Examples

Updated: 2013-10-25

Package command overview

MultivariateStirlingP1[*n*, *k*]

returns the multivariate Stirling polynomial $S_{n,k}$ of the first kind in $n-k+1$ indeterminates

MultivariateStirlingA[*n*, *k*]

returns the rational function $A_{n,k}$ defined by $X[1]^{-(2n-1)} \times$ MultivariateStirlingP1[n, k]

MultivariateStirlingP2[*n*, *k*]

returns the multivariate Stirling polynomial $B_{n,k}$ of the second kind in $n-k+1$ indeterminates (= partial Bell polynomial)

SetVariablesTo[{*var1*, *var2*, ...}]

generates a rule set that converts indeterminates $X[1]$, $X[2]$, ... into *var1*, *var2*, ...

SubIndexed[*m*]

generates a special rule set that converts X[1],...,X[m] into $X_1,\ldots,X_m$

AssociateBellPolynomial[*n*, *k*]

returns the partial Bell polynomial $B_{n,k}(0, X[2], \ldots, X[n+k+1])$ with 0 substituted in place of $X[1]$

LahPolynomial[*n*, *k*]

returns the multivariate Lah polynomial $L_{n,k}$ in $n-k+1$ indeterminates

CauchyPolynomial[*n*, *k*]

returns the multivariate Cauchy polynomial $C_{n,k}$ in $n-k+1$ indeterminates

■ Warning

The symbol X is used as the basis letter denoting indeterminates; it is protected within this package, that is, you cannot change its value:

```
X = 4;
```

Set::wrsym : Symbol X is Protected. ≫

However ...

```
{X[1], X[2], X[3]} /. SetVariablesTo[{-5, 7}]
```

```
{-5, 7, X[3]}
```

Read in the package file

In order to evaluate the cell below, both files "MultivariateStirlingPolynomials.m" and "MultivariateStirlingPolynomialsExamples.nb" must have been copied into your working directory.

```
SetDirectory[NotebookDirectory[]];
<< MultivariateStirlingPolynomials`
```

Something new about a classical topic

The multivariate Stirling polynomials of the first kind—as I would like to call this new class of polynomials—are closely connected to the well-known Bell polynomials. This became clear to me when I studied higher Lie derivatives of scalar functions and Faà di Bruno's chain rule.

"It would be surprising if anything new could be said about such a classical topic ..."

Huang / Marcantognini / Young: Chain Rules for Higher Derivatives.
The Mathematical Intelligencer 28/2 (2006)

Generate Stirling polynomials of the second kind

Let's start with some well-known stuff.
Multivariate Stirling Polynomials (MSPs) of the second kind are the same as partial Bell Polynomials $(B_{n,k})$.
Here comes the Bell polynomial $B_{6,4}$:

```
MultivariateStirlingP2[6, 4]
```

$45\, X[1]^2\, X[2]^2 + 20\, X[1]^3\, X[3]$

If you don't like the indeterminates notated as $X[1]$, $X[2]$, ..., try this:

```
MultivariateStirlingP2[6, 4] /. SetVariablesTo[{x, y, z}]
```

$45\, x^2\, y^2 + 20\, x^3\, z$

or that:

```
MultivariateStirlingP2[6, 4] /. SubIndexed[6 - 4 + 1]
```

$45\, X_1^2\, X_2^2 + 20\, X_1^3\, X_3$

Replacing all indeterminates by 1, gives the sum of the coefficients:

```
MultivariateStirlingP2[6, 4] /. SetVariablesTo[{1, 1, 1}]
```

65

Recall that this is a Stirling number of the second kind:

```
StirlingS2[6, 4]
```

65

Finally, let's create a nice triangular matrix of partial Bell polynomials:

```
BMatrix = Table[Table[MultivariateStirlingP2[i, j], {j, 1, 4}], {i, 1, 4}];
BMatrix /. SubIndexed[4] // MatrixForm
```

$$\begin{pmatrix} X_1 & 0 & 0 & 0 \\ X_2 & X_1^2 & 0 & 0 \\ X_3 & 3 X_1 X_2 & X_1^3 & 0 \\ X_4 & 3 X_2^2 + 4 X_1 X_3 & 6 X_1^2 X_2 & X_1^4 \end{pmatrix}$$

Generate Stirling polynomials of the first kind

The polynomial family $S_{n,k}$, $1 \leq k \leq n$, is —as a whole— new.
Here comes their 5-th generation consisting of the members $S_{5,i}$ $(1 \leq i \leq 5)$:

```
Table[MultivariateStirlingP1[5, i], {i, 1, 5}] // TableForm
```

```
105 X[2]^4 - 105 X[1] X[2]^2 X[3] + 10 X[1]^2 X[3]^2 + 15 X[1]^2 X[2] X[4] - X[1]^3 X[5]
-105 X[1] X[2]^3 + 60 X[1]^2 X[2] X[3] - 5 X[1]^3 X[4]
45 X[1]^2 X[2]^2 - 10 X[1]^3 X[3]
-10 X[1]^3 X[2]
X[1]^4
```

Replacing every $X[j]$ by 1, again yields Stirling numbers:

```
% /. SetVariablesTo[{1, 1, 1, 1, 1}]
```

```
{24, -50, 35, -10, 1}
```

These, however, are **signed** Stirling numbers of the **first** kind:

```
Table[StirlingS1[5, i], {i, 1, 5}]
```

```
{24, -50, 35, -10, 1}
```

Now, let's create an SMatrix analogous to the preceding BMatrix:

```
SMatrix = Table[Table[MultivariateStirlingP1[i, j], {j, 1, 4}], {i, 1, 4}];
SMatrix /. SubIndexed[4] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ -X_2 & X_1 & 0 & 0 \\ 3 X_2^2 - X_1 X_3 & -3 X_1 X_2 & X_1^2 & 0 \\ -15 X_2^3 + 10 X_1 X_2 X_3 - X_1^2 X_4 & 15 X_1 X_2^2 - 4 X_1^2 X_3 & -6 X_1^2 X_2 & X_1^3 \end{pmatrix}$$

A fundamental law of inversion

The product of both matrices comes as a surprise:

```
SMatrix.BMatrix // Simplify // MatrixForm
```

$$\begin{pmatrix} X[1] & 0 & 0 & 0 \\ 0 & X[1]^3 & 0 & 0 \\ 0 & 0 & X[1]^5 & 0 \\ 0 & 0 & 0 & X[1]^7 \end{pmatrix}$$

This gives evidence to the fact that $A_{n,k} := X_1^{-(2n-1)} S_{n,k}$ and $B_{n,k}$ meet a condition strongly generalizing the well-known inversion law of the Stirling numbers of the first and second kind:

```
AMatrix = Table[Table[MultivariateStirlingA[i, j], {j, 1, 4}], {i, 1, 4}];
AMatrix // MatrixForm
```

$$\begin{pmatrix} \frac{1}{X[1]} & 0 & 0 & 0 \\ -\frac{X[2]}{X[1]^3} & \frac{1}{X[1]^2} & 0 & 0 \\ \frac{3X[2]^2}{X[1]^5} - \frac{X[3]}{X[1]^4} & -\frac{3X[2]}{X[1]^4} & \frac{1}{X[1]^3} & 0 \\ -\frac{15X[2]^3}{X[1]^7} + \frac{10X[2]X[3]}{X[1]^6} - \frac{X[4]}{X[1]^5} & \frac{15X[2]^2}{X[1]^6} - \frac{4X[3]}{X[1]^5} & -\frac{6X[2]}{X[1]^5} & \frac{1}{X[1]^4} \end{pmatrix}$$

Then, the **inversion law for multivariate Stirling polynomials** is as follows:

```
AMatrix.BMatrix // Expand // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Of course, also the following holds:

```
BMatrix.AMatrix // Expand // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

The main result

Theorem 6.1 in my paper on *Multivariate Stirling Polynomials of the First and Second Kind* (to appear) states that for all $n \geq k \geq 1$ the following equation holds:

$$S_{n,k} = \sum_{r=k-1}^{n-1} (-1)^{n-1-r} \binom{2n-2-r}{k-1} X_1^r B_{2n-1-k-r,n-1-r}(0, X[2], \ldots, X[n-k+1])$$

Let's try an instance:

```
n = 7; k = 3;
MultivariateStirlingP1[n, k] /. SubIndexed[n - k + 1]
Sum[(-1)^(n-1-r) Binomial[2 n - 2 - r, k - 1] X[1]^r AssociateBellPolynomial[2 n - 1 - k - r, n - 1 - r],
   {r, k - 1, n - 1}] /. SubIndexed[n - k + 1] // Expand
```

$4725 X_1^2 X_2^4 - 3780 X_1^3 X_2^2 X_3 + 280 X_1^4 X_3^2 + 420 X_1^4 X_2 X_4 - 21 X_1^5 X_5$

$4725 X_1^2 X_2^4 - 3780 X_1^3 X_2^2 X_3 + 280 X_1^4 X_3^2 + 420 X_1^4 X_2 X_4 - 21 X_1^5 X_5$

■ Lagrange inversion

In the **special case** $k = 1$ we get the remarkable result that $S_{n,1}$ can be used to invert a power series $p(x) = a_1 x + a_2 x^2 + a_3 x^3 + \ldots$ $(a_1 \neq 0)$. More precisely: Let $b_n := a_1^{-(2n-1)} S_{n,1}(a_1, \ldots, a_n)$. Then, b_n is the n-th coefficient of the inverse of $p(x)$, that is, we have $p^{-1}(x) = b_1 x + b_2 x^2 + b_3 x^3 + \ldots$, where $p(p^{-1}(x)) = p^{-1}(p(x)) = x$.

Relatives of the Bell polynomials

Whenever we replace the indeterminates $X[j]$ in $B_{n,k}$ by a multiplum $c_j X[j]$ (for all $j = 1, 2, \ldots, n-k+1$), we obtain a polynomial closely related to the original Bell polynomials. Let's call it a **relative of** $B_{n,k}$.

Cauchy polynomials

For $c_j = (j-1)!$ the result looks like this:

```
CauchyPolynomial[7, 4] /. SubIndexed[7 - 4 + 1]
```

$105\, X_1 X_2^3 + 420\, X_1^2 X_2 X_3 + 210\, X_1^3 X_4$

Why "Cauchy"?
Have, for instance, a look at the first coefficient! 105 is the number of permutations having 1 cycle of length 1 and 3 cycles of length 2. This condition is mirrored by the monomial $X_1 X_2^3$. Cauchy has found a famous expression that computes these numbers. Of course, the sum of all these counts the number of permutations (here: of 7 elements) consisting of 4 cycles. This is the **signless** Stirling number of the first kind $c(7, 4)$:

```
CauchyPolynomial[7, 4] /. SetVariablesTo[{1, 1, 1, 1}]
```

735

```
StirlingS1[7, 4]
```

-735

Lah polynomials

Counting linearly ordered subsets (blocks or parts of a partition) instead of cycles, gives the Lah numbers (named after Ivo Lah) as coefficients. The resulting polynomials may be called Lah polynomials. Here $c_j = j!$ for $j = 1, 2, \ldots, n-k+1$.

```
LahPolynomial[6, 2]
```

$360\, X[3]^2 + 720\, X[2]\, X[4] + 720\, X[1]\, X[5]$

Consider the sum of all coefficients, that is: the **signless** Lah number corresponding to this polynomial:

```
LahPolynomial[6, 2] /. SetVariablesTo[Table[1, {5}]]
```

1800

This result is the number of ways a set of $n = 6$ elements can be partitioned into $k = 2$ nonempty linearly ordered subsets. It can be simply expressed by the combinatorial term: $\frac{n!}{k!}\binom{n-1}{k-1}$.

```
n = 6; k = 2;
n! * Binomial[n - 1, k - 1] / k!
```

1800

Space for your experiments